use "${clean}newspaper_analysis.dta", clear

replace north = 1 if state == "Missouri" ///
	| state == "Missouri Territory" ///
	| state == "Maryland" ///
	| state == "Delaware" ///
	| state == "Kentucky" ///
	| state == "West Virginia" ///

keep if north == 1 | south == 1

// recoding Alexandria, VA to the South (technically part of DC until 1844)
replace south = 1 if city == "Alexandria"
replace north = 0 if city == "Alexandria"
replace state = "Virginia" if city == "Alexandria"


// standardizing variables
egen county_surban = std(county_urban_percent)
egen county_srugged = std(county_ruggedness)
egen county_spostoffice = std(county_postoffice_area)

// keeping the sample consistent
cap drop sample
gen sample = 1
foreach var of varlist $county_npcontrols {
	replace sample = . if `var' == .
}

// creating some variables
local y = 1860

gen period = year > `y'
replace year = year - `y'

	
* regressions
* ------------

xtset pub_id

estimates clear	

//  re model
eststo: xtreg singular c.year##i.period##i.north ${county_npcontrols} if sample == 1, re cluster(pub_id)
estadd local fe "No"
	
//  fe model
eststo: xtreg singular c.year##i.period##i.north ${county_npcontrols} if sample == 1, fe cluster(pub_id)
estadd local fe "Publication"


// table
esttab est* using "${output}TabD7_np_northborder_`y'_fulltable.tex", replace ///
	b(%8.3f) se(%8.3f) ///
	starlevels(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
	scalars("fe Fixed Effects") ///
	keep( ///
		year ///
		1.period ///
		1.north ///
		1.period#c.year ///
		1.north#c.year ///
		1.period#1.north ///
		1.period#1.north#c.year ///
		county_surban county_spostoffice county_srugged county_anyriver county_anycanal _cons ///
		) ///
	order( ///
		year ///
		1.period ///
		1.north ///
		1.period#c.year ///
		1.north#c.year ///
		1.period#1.north ///
		1.period#1.north#c.year ///
		county_surban county_spostoffice county_srugged county_anyriver county_anycanal _cons ///
		) ///
	coeflabels( ///
		year "Year" ///
		1.period "After `y'" ///
		1.north "North" ///
		1.period#c.year "Year $\times$ After `y'" ///
		1.north#c.year "Year $\times$ North" ///
		1.period#1.north "After `y' $\times$ North" ///
		1.period#1.north#c.year "Year $\times$ After `y' $\times$ North" ///
		county_surban "County \% Urban (std)" ///
		county_spostoffice "Post Office Density (std)" ///
		county_srugged "Terrain Ruggedness (std)" ///
		county_anyriver "County on River" ///
		county_anycanal "County on Canal" ///
		_cons "Constant" ///
		) ///
	nodepvars nomtitles nonotes ///
	nogaps nolines posthead(\midrule) prefoot(\midrule) postfoot("\bottomrule \end{tabular}}") ///
	tex compress	



